MEMORY
{
PAGE 0 :
   #if BOOT_FROM_FLASH
      /* BEGIN is used for the "boot to Flash" bootloader mode   */
      BEGIN           : origin = 0x080000, length = 0x000002
   #else
      /* BEGIN is used for the "boot to SARAM" bootloader mode   */
      BEGIN           : origin = 0x000000, length = 0x000002
   #endif // BOOT_FROM_FLASH
   
   #ifdef CLA_BLOCK_INCLUDED
   /* Need to set appropriate registers to access the RAM as CLA program memory */
        #if (CPU_RAMLS_PROG_LENGTH > 0)
        RAMLS_PROG          : origin = CPU_RAMLS_PROG_START, length = CPU_RAMLS_PROG_LENGTH
        #endif //CPU_RAMLS_PROG_LENGTH
        RAMLS_CLA_PROG      : origin = CLA_RAMLS_PROG_START, length = CLA_RAMLS_PROG_LENGTH
        /* Need to set appropriate registers to access the RAM as CLA data memory */
        RAMLS_CLA_DATA      : origin = CLA_RAMLS_DATA_START, length = CLA_RAMLS_DATA_LENGTH
   #else
        RAMLS_PROG          : origin = 0x00A800, length = 0x001800
   #endif //CLA_BLOCK_INCLUDED
 
   //RAMGS_PROG         : origin = 0x00C000, length = 0x04000 /* RAMGS0 and RAMGS1 - Program */
 
   RESET              : origin = 0x3FFFC0, length = 0x000002
   
   /* Flash sectors */
   /* BANK 0 */
   FLASH_BANK0_SEC0   : origin = 0x080002, length = 0x1FFFE    /* on-chip Flash */
   /* BANK 2 */
   FLASH_BANK2        : origin = 0x0C0000, length = 0x1FFFF    /* on-chip Flash */
 
   BOOTROM            : origin = 0x003F8000, length = 0x00007FC0
   
PAGE 1 :
 
   BOOT_RSVD          : origin = 0x000002, length = 0x0001BE
   RAMM0M1            : origin = 0x0001C0, length = 0x000640     /* on-chip RAM block M0 and M1 */
 
   #ifndef CLA_BLOCK_INCLUDED  
        RAMLS_DATA    : origin = 0x008000, length = 0x002800
   #endif //CLA_BLOCK_INCLUDED
 
   RAMLS_DATA89       : origin = 0x014000, length = 0x04000 /* RAMLS8 and RAMLS9 - DATA */
   RAMGS_DATA         : origin = 0x00C000, length = 0x04000 /* RAMGS0 and RAMGS1 - DATA */
 
   CLA1_MSGRAMLOW     : origin = 0x001480, length = 0x000080
   CLA1_MSGRAMHIGH    : origin = 0x001500, length = 0x000080
 
   CLATODMARAM        : origin = 0x001680,   length = 0x000080
   DMATOCLARAM        : origin = 0x001700,   length = 0x000080
}
 
SECTIONS
{
   #if BOOT_FROM_FLASH
      codestart        : > BEGIN,               PAGE = 0, ALIGN(8)
      .text            : > FLASH_BANK0_SEC0 | FLASH_BANK2,      PAGE = 0, ALIGN(8)      
      .cinit           : > FLASH_BANK0_SEC0,      PAGE = 0, ALIGN(8)
      .switch          : > FLASH_BANK0_SEC0,      PAGE = 0, ALIGN(8)
      .reset           : > RESET,               PAGE = 0, TYPE = DSECT /* not used, */
      .stack           : > RAMM0M1,             PAGE = 1
   
      #if defined(__TI_EABI__)
        .init_array    : > FLASH_BANK0_SEC0,      PAGE = 0, ALIGN(8)
        .bss           : > RAMLS_DATA89,          PAGE = 1
        .bss:output    : > RAMLS_DATA89,          PAGE = 1
        .bss:cio       : > RAMLS_DATA89,          PAGE = 1
        .const         : > FLASH_BANK0_SEC0,      PAGE = 0, ALIGN(8)
        .data          : > RAMLS_DATA89,          PAGE = 1
        .sysmem        : > RAMLS_DATA89,          PAGE = 1  
        .TI.ramfunc : {} LOAD = FLASH_BANK0_SEC0,
                         RUN =  RAMLS_PROG,
                         LOAD_START(RamfuncsLoadStart),
                         LOAD_SIZE(RamfuncsLoadSize),
                         LOAD_END(RamfuncsLoadEnd),
                         RUN_START(RamfuncsRunStart),
                         RUN_SIZE(RamfuncsRunSize),
                         RUN_END(RamfuncsRunEnd),
                         PAGE = 0, ALIGN(8)
     
        ramfuncs       : LOAD = FLASH_BANK0_SEC0,
                         RUN =  RAMLS_PROG,
                         LOAD_START(MW_RamfuncsLoadStart),
                         LOAD_SIZE(MW_RamfuncsLoadSize),
                         LOAD_END(MW_RamfuncsLoadEnd),
                         RUN_START(MW_RamfuncsRunStart),
                         RUN_SIZE(MW_RamfuncsRunSize),
                         RUN_END(MW_RamfuncsRunEnd),
                         PAGE = 0, ALIGN(8)
 
        #ifdef CLA_BLOCK_INCLUDED
          /* CLA specific sections */
            Cla1Prog   : LOAD = FLASH_BANK0_SEC0,
                         RUN = RAMLS_CLA_PROG,
                         LOAD_START(Cla1ProgLoadStart),
                         RUN_START(Cla1ProgRunStart),
                         LOAD_SIZE(Cla1ProgLoadSize),
                         PAGE = 0, ALIGN(4)
            .const_cla : LOAD = FLASH_BANK0_SEC0,
                         RUN = RAMLS_CLA_DATA,
                         RUN_START(Cla1ConstRunStart),
                         LOAD_START(Cla1ConstLoadStart),
                         LOAD_SIZE(Cla1ConstLoadSize),
                         PAGE = 0
        #endif //CLA_BLOCK_INCLUDED
 
      #else
        .pinit         : > FLASH_BANK0_SEC0,     PAGE = 0, ALIGN(4)
        .econst        : > FLASH_BANK0_SEC0,     PAGE = 0, ALIGN(4)
        .cio           : > RAMLS_DATA89,         PAGE = 1
        .ebss          : > RAMLS_DATA89,         PAGE = 1
        .esysmem       : > RAMLS_DATA89,         PAGE = 1  
        .TI.ramfunc : {} LOAD = FLASH_BANK0_SEC0,
                         RUN =  RAMLS_PROG,
                         LOAD_START(_RamfuncsLoadStart),
                         LOAD_SIZE(_RamfuncsLoadSize),
                         LOAD_END(_RamfuncsLoadEnd),
                         RUN_START(_RamfuncsRunStart),
                         RUN_SIZE(_RamfuncsRunSize),
                         RUN_END(_RamfuncsRunEnd),
                         PAGE = 0, ALIGN(8)
     
        ramfuncs       : LOAD = FLASH_BANK0_SEC0,
                         RUN =  RAMLS_PROG,
                         LOAD_START(_MW_RamfuncsLoadStart),
                         LOAD_SIZE(_MW_RamfuncsLoadSize),
                         LOAD_END(_MW_RamfuncsLoadEnd),
                         RUN_START(_MW_RamfuncsRunStart),
                         RUN_SIZE(_MW_RamfuncsRunSize),
                         RUN_END(_MW_RamfuncsRunEnd),
                         PAGE = 0, ALIGN(8)
 
        #ifdef CLA_BLOCK_INCLUDED
          /* CLA specific sections */
            Cla1Prog   : LOAD = FLASH_BANK0_SEC0,
                         RUN = RAMLS_CLA_PROG,
                         LOAD_START(_Cla1ProgLoadStart),
                         RUN_START(_Cla1ProgRunStart),
                         LOAD_SIZE(_Cla1ProgLoadSize),
                         PAGE = 0, ALIGN(4)
            .const_cla : LOAD = FLASH_BANK0_SEC0,
                         RUN = RAMLS_CLA_DATA,
                         RUN_START(_Cla1ConstRunStart),
                         LOAD_START(_Cla1ConstLoadStart),
                         LOAD_SIZE(_Cla1ConstLoadSize),
                         PAGE = 0
        #endif //CLA_BLOCK_INCLUDED
 
      #endif //defined(__TI_EABI__)
     
      ramgs0           : > RAMLS_DATA89,          PAGE = 1
      ramgs1           : > RAMLS_DATA89,          PAGE = 1  
 
    /* Allocate IQ math areas: */
      IQmath           : > FLASH_BANK0_SEC0,     PAGE = 0, ALIGN(8)            /* Math Code */
      IQmathTables     : > FLASH_BANK0_SEC0,     PAGE = 0, ALIGN(8)
 
   #else
      codestart        : > BEGIN,               PAGE = 0
      .TI.ramfunc      : > RAMLS_PROG,          PAGE = 0
      ramfuncs         : > RAMLS_PROG,          PAGE = 0
      .text            : >> RAMLS_DATA89 | RAMGS_DATA | RAMLS_DATA, PAGE = 1
      .cinit           : > RAMGS_DATA,          PAGE = 1
      .switch          : >> RAMGS_DATA | RAMLS_DATA | RAMLS_DATA89, PAGE = 1
      .reset           : > RESET,               PAGE = 0, TYPE = DSECT /* not used, */
      .stack           : > RAMM0M1,             PAGE = 1
 
      #if defined(__TI_EABI__)
        .bss           : > RAMLS_DATA89,          PAGE = 1
        .bss:output    : > RAMLS_DATA89,          PAGE = 1
        .bss:cio       : > RAMLS_DATA89,          PAGE = 1
        .init_array    : > RAMM0M1,             PAGE = 1
        .const         : >> RAMLS_DATA89 | RAMGS_DATA, PAGE = 1
        .data          : >  RAMLS_DATA89,              PAGE = 1
        .sysmem        : >> RAMLS_DATA89 | RAMGS_DATA, PAGE = 1
      #else
        .pinit         : >  RAMM0M1,               PAGE = 1
        .cio           : >> RAMGS_DATA | RAMLS_DATA89, PAGE = 1
        .ebss          : >> RAMLS_DATA89 | RAMGS_DATA, PAGE = 1
        .econst        : >> RAMGS_DATA | RAMLS_DATA89, PAGE = 1
        .esysmem       : >> RAMGS_DATA | RAMLS_DATA89, PAGE = 1
      #endif //defined(__TI_EABI__)
 
      ramgs0           : > RAMLS_DATA89,          PAGE = 1
      ramgs1           : > RAMLS_DATA89,          PAGE = 1
 
      /* The following section definition are for SDFM examples */
      Filter1_RegsFile : > RAMLS_PROG,          PAGE = 0
      Filter2_RegsFile : > RAMLS_PROG,          PAGE = 0
      Filter3_RegsFile : > RAMLS_PROG,          PAGE = 0
      Filter4_RegsFile : > RAMLS_PROG,          PAGE = 0
   
      /* Allocate IQ math areas: */
      IQmath           : > RAMLS_PROG,          PAGE = 0            /* Math Code */
      IQmathTables     : > RAMLS_PROG,          PAGE = 0
 
      #ifdef CLA_BLOCK_INCLUDED
          /* CLA specific sections */
          Cla1Prog      : > RAMLS_CLA_PROG,     PAGE = 0
          .const_cla    : > RAMLS_CLA_DATA,     PAGE = 0
      #endif //CLA_BLOCK_INCLUDED        
   #endif //BOOT_FROM_FLASH
 
   #ifdef CLA_BLOCK_INCLUDED
        // CLA Sections
        // Must be allocated to memory the CLA has write access to
        .scratchpad     : > RAMLS_CLA_DATA,     PAGE = 0
        .bss_cla        : > RAMLS_CLA_DATA,     PAGE = 0
        Cla1ToCpuMsgRAM : > CLA1_MSGRAMLOW,     PAGE = 1
        CpuToCla1MsgRAM : > CLA1_MSGRAMHIGH,    PAGE = 1
        Cla1DataRam0    : > RAMLS_CLA_DATA,     PAGE = 0
   #endif //CLA_BLOCK_INCLUDED
 
   //xcpTl_buf : > RAMLS_DATA89, PAGE = 1   /* move XCP buffer to global RAM */
   //.xcp      : > RAMGS_DATA, PAGE = 1  
}
 
/*
//===========================================================================
// End of file.
//===========================================================================
*/
 
